

一个依值类型 Lambda 演算的教学实现

Andres Löb

原作者
乌特勒支大学

Conor McBride

原作者
斯特拉斯克莱德大学

Wouter Swierstra

原作者
诺丁汉大学

Chuigda Whitegive

翻译
第七通用设计局

Gemini

类型论支持、校对
Google Deepmind

Claude

校对
Anthropic

译者前言

本文是对文章 [A tutorial implementation of a dependently typed lambda calculus](#) 的中文翻译，部分字句有所改动。排版经过调整，使图表总是位于直接描述该图表的文本下方，以便阅读。术语 (*terminology*) 在正文中第一次出现的地方以仿宋体（中文）或 *Italic (English)* 呈现，如果某个术语难以辨认，则总是会以这种形式呈现。本文附有[其它语言的实现](#)。

摘要

本文介绍了依值类型演算核心的类型规则，并提供了一个简洁的 Haskell 实现。本文着重阐述了从简单类型 Lambda 演算过渡到依值类型 Lambda 演算所需的变更。此外，本文还描述了如何扩展核心语言的数据类型，并给出了几个小型示例程序。本文附带一个[可执行解释器和示例代码](#)，方便读者立即体验本文所描述的系统。

1. 介绍

许多函数式程序员都对使用依值类型 (*dependent type*) 编程犹豫不决。他们常说：依值类型会使类型检查变得不可判定，类型检查器总是陷入死循环，以及依值类型非常、非常难。

然而，同一批程序员却非常热衷于使用各种复杂的类型系统扩展进行编程。比如说，现在的 Haskell 实现支持广义代数数据类型，带函数依赖的多参数类型类，关联类型和类型家族，非直谓 (*impredicative*) 高阶类型，凡此种种。程序员们似乎对依值类型唯恐避之不及。

函数式社区对依值类型普遍缺乏理解，这是阻碍依值类型进一步普及的主要障碍之一。尽管目前已经有了不少基于依值类型的优秀实验工具和编程语言，理解这些工具到底是如何工作的却是一件难事。很大一部分关于依值类型的文献都是由类型学家撰写、供其他类型学家阅读的，这些文献对函数式程序员来说并不友好。本文致力于弥补这一现状。

本文从简单类型 Lambda 演算 (*simply-typed lambda calculus*) (第二节) 开始，给出了抽象语法、求值和类型检查的数学规范和 Haskell 实现。以简单类型 Lambda 演算为起点，本文进一步地研究最小化的依值类型 Lambda 演算 (*dependently typed lambda calculus*) (第三节)。

受 Pierce 逐步发展类型系统[21]的启发，本文重点阐述了向依值类型 Lambda 演算过渡所需的规范和实现方面的变更。令人惊讶的是，所需的变更并没有想象中那么多。我们希望通过尽可能明确地阐述这些变更，能让已经熟悉简单类型 Lambda 演算的读者尽可能顺利地过渡到依值类型。

尽管本文中没有发明新的类型系统，我们相信本文可以作为 Haskell 中依值类型系统实现的入门教程。实现一个类型系统是学习其中所有微妙细节的最佳途径之一。尽管我们不打算全面探讨所有实现类型化 Lambda 演算的方法，但我们会尽量明确阐述我们的设计决策，仔细提供其他选择，并概述更广泛的设计空间。

只有将数据类型添加到这个基础演算中，才能真正发挥依值类型的全部威力。因此，我们在第四节中演示了如何使用自然数和向量 (*vector*) 扩展我们的语言。使用这些新增的数据类型，我们编写了经典的向量追加操作，以演示如何在我们的核心演算中进行编程。而利用本节解释的原理，可以向演算中加入更多数据类型。

最后，我们简化了系统实验流程：本文源代码包含一个小型解释器，用于解释我们描述的类型系统和求值规则。由于使用了与本文相同的源代码，该解释器能够保证严格遵循我们描述的实现，并且文档齐全。因此，它为进一步学习和实验提供了一个宝贵的平台。

本文并非对依值类型编程的入门介绍，也未讲解如何实现完整的依值类型编程语言。然而，我们希望本文能够消除函数式程序员对依值类型的诸多误解，并鼓励读者进一步探索这一令人兴奋的研究领域。

2. 简单类型 Lambda 演算

在探索依值类型的过程中，我们希望从熟悉的领域入手。因此，在本节中，我们将讨论简单类型 Lambda 演算，简称 λ_{\rightarrow} 。从某种意义上来说， λ_{\rightarrow} 是最小的静态类型函数式语言。每个词项 (*term*)¹ 都显式地标注了类型，不需要类型推理。相比于作为 ML 或 Haskell 等支持多态类型和类型构造子 (*type constructor*) 的语言的基础的那些类型化 Lambda 演算 (*typed lambda calculi*)， λ_{\rightarrow} 的结构要简单得多。在 λ_{\rightarrow} 中只有基础类型，且函数类型不能是多态的 (*polymorphic*)。如果不增加额外的规则， λ_{\rightarrow} 是强正规化 (*strongly normalizing*) 的：对于任何词项，无论求值策略如何，求值总是能停机。

2.1. 抽象语法

λ_{\rightarrow} 的类型语言仅由两种结构组成。存在一组基本类型 α ；复合类型 $\tau \rightarrow \tau'$ 则对应着接受一个 τ ，返回一个 τ' 的函数：

$$\begin{array}{ll} \tau ::= \alpha & \text{基本类型} \\ | \tau \rightarrow \tau' & \text{函数类型} \end{array}$$

λ_{\rightarrow} 共有四种词项：带显式类型注解 (*type annotation*) 的词项、变量 (*variable*)、应用 (*application*) 和 Lambda 抽象 (*lambda abstraction*)：

$$\begin{array}{ll} e ::= e :: \tau & \text{带类型注解的词项}^2 \\ | x & \text{变量} \\ | e e' & \text{应用} \\ | \lambda x \rightarrow e & \text{Lambda 抽象} \end{array}$$

词项可以被求值为值 (*value*)。一个值要么是一个中性项 (*neutral term*) —— 一个被应用了零个或多个值的变量，要么是一个 Lambda 抽象：

$$\begin{array}{ll} v ::= n & \text{中性项} \\ | \lambda x \rightarrow v & \text{Lambda 抽象} \\ n ::= x & \text{变量} \\ | n v & \text{应用} \end{array}$$

¹译注：词项这一术语翻译取自练琪灏的依值类型入门讲义：<https://m.or.gd/notes/mltt-20230703.pdf>。

²类型学家使用符号 “ \vdash ” 或者 “ \in ” 表示类型归属关系。然而，在 Haskell 中，符号 “ \vdash ” 被用作列表的 `cons` 运算符，因此 Haskell 的设计者为类型注解选择了非标准的 “ $::$ ”。本文将尽可能地遵循 Haskell 语法，以缩小本文所涉及的不同语言之间的语法差距。

2.2. 求值

λ_{\rightarrow} 的求值规则（大步）如图 1 所示。记号 $e \Downarrow v$ 表示将 e 完全求值的结果是 v 。因为 λ_{\rightarrow} 是强正规化的语言，求值策略自然无关紧要。简单起见，我们尽可能地把所有东西求值到底——包括 Lambda 内部的东西。类型注解在求值阶段会被忽略。求值变量会得到变量自身。

唯一值得关注的情况是应用：在对应用求值时，需要依据应用左边的子项的求值结果作出不同的行动：当左子项的求值结果是中性项时，求值无法继续进行，此时我们把两个子项的求值结果组合成一个新的中性项；当左子项的求值结果是 Lambda 抽象时，我们进行 β -归约 (β -reduce)，这一过程可能产生新的可归约项 (*redex*, *reducible-expression*, 直译为可归约表达式)，因此我们还要对 β -归约的结果继续求值。

$$\begin{array}{c}
 \frac{e \Downarrow v}{e :: \tau \Downarrow v} \quad \text{类型注解会被忽略}^3 \\
 \\
 \frac{}{x \Downarrow x} \quad \text{求值变量会得到变量自身} \\
 \\
 \frac{e \Downarrow v}{\lambda x \rightarrow e \Downarrow \lambda x \rightarrow v} \quad \text{求值 Lambda 抽象的内部} \\
 \\
 \frac{e \Downarrow \lambda x \rightarrow v \quad v[x \mapsto e'] \Downarrow v'}{e e' \Downarrow v'} \quad \text{对应用求值：左边的子项是 Lambda 抽象} \\
 \\
 \frac{e \Downarrow n \quad e' \Downarrow v'}{e e' \Downarrow n v'} \quad \text{对应用求值：左边的子项是中性项}
 \end{array}$$

图 1 λ_{\rightarrow} 的求值规则

作为例子，以下是 λ_{\rightarrow} 中的一些词项和它们的求值结果。我们用 `id` 代指词项 $\lambda x \rightarrow x$ ，用 `const` 代指词项 $\lambda x y \rightarrow x$ ——这种写法是 $\lambda x \rightarrow \lambda y \rightarrow x$ 的语法糖。

$$\begin{array}{l}
 (\text{id} :: \alpha \rightarrow \alpha) y \Downarrow y \\
 (\text{const} :: (\beta \rightarrow \beta) \rightarrow \alpha \rightarrow \beta \rightarrow \beta) \text{id } y \Downarrow \text{id}
 \end{array}$$

2.3. 类型系统

类型规则 (*type rule*) 大都形如 $\Gamma \vdash e :: \tau$ ，这表示在上下文 Γ 中，词项 e 的类型是 τ 。类型上下文中列出了有效的基本类型，并将标识符与类型信息相关联。我们用 $\alpha :: *$ 表示 α 是一个基本类型，用 $x :: \tau$ 表示词项 x 的类型是 τ 。

$$\begin{array}{ll}
 \Gamma ::= \varepsilon & \text{空的类型上下文} \\
 | \Gamma, \alpha :: * & \text{添加一个类型标识符} \\
 | \Gamma, x :: \tau & \text{添加一个词项标识符}
 \end{array}$$

词项和类型中的每个自由变量 (*free variable*) 都必须在类型上下文中出现。例如，如果我们要声明 `const` 具有类型 $(\beta \rightarrow \beta) \rightarrow \alpha \rightarrow \beta \rightarrow \beta$ ，则类型上下文中至少应该包含：

$$\alpha :: *, \beta :: *, \text{const} :: (\beta \rightarrow \beta) \rightarrow \alpha \rightarrow \beta \rightarrow \beta$$

注意 α 和 β 是在它们被 `const` 的类型使用前引入的。

³译注：这些说明性文本是译者加的。

基于上述规则，我们在图 2 中给出了类型上下文及其有效性的定义。

$$\frac{}{\text{valid}(\varepsilon)} \quad \frac{\text{valid}(\Gamma)}{\text{valid}(\Gamma, \alpha :: *)} \quad \frac{\text{valid}(\Gamma) \quad \Gamma \vdash \tau :: *}{\text{valid}(\Gamma, x :: \tau)}$$

$$\frac{\Gamma(\alpha) = *}{\Gamma \vdash \alpha :: *} \text{ [TVAR]} \quad \frac{\Gamma \vdash \tau :: * \quad \Gamma \vdash \tau' :: *}{\Gamma \vdash \tau \rightarrow \tau' :: *} \text{ [FUN]}$$

图 2 λ_{\rightarrow} 中的类型上下文和良构 (*well-formed*) 的类型

图 2 中的最后两条规则解释了一个类型在什么情况下是良构的——当类型中的所有自由变量都在上下文中的时候。在类型良构性 (*well-formedness*) 规则以及后续的类型规则中，我们隐含地假设所有类型上下文都有效。

注意 λ_{\rightarrow} 不是多态的：每个类型标识符都代表一个具体的类型，不能被实例化。

最终，我们可以给出如图 3 所示的类型规则。我们不推断被 Lambda 绑定的变量的类型。也就是说，总体上而言，我们只进行类型检查。不过我们可以很容易地判定带类型注解的项、变量和应用的类型。因此，当一个类型是类型检查算法的输入时，我们给类型规则标上 $::_{\downarrow}$ ；当类型是类型检查算法的输出时，我们给类型规则标上 $::_{\uparrow}$ 。目前这些记号只是为了提供一些直觉，但在实现中，两种规则之间的差异会变得明显。

$$\frac{\Gamma \vdash \tau :: * \quad \Gamma \vdash e ::_{\downarrow} \tau}{\Gamma \vdash (e :: \tau) ::_{\uparrow} \tau} \text{ [ANN]} \quad \frac{\Gamma(x) = \tau}{\Gamma \vdash x ::_{\downarrow} \tau} \text{ [VAR]}$$

$$\frac{\Gamma \vdash e ::_{\uparrow} \tau \rightarrow \tau' \quad \Gamma \vdash e' ::_{\downarrow} \tau}{\Gamma \vdash e e' ::_{\downarrow} \tau'} \text{ [APP]}$$

$$\frac{\Gamma \vdash e ::_{\uparrow} \tau}{\Gamma \vdash e ::_{\downarrow} \tau} \text{ [CHK]} \quad \frac{\Gamma, x :: \tau \vdash e ::_{\downarrow} \tau'}{\Gamma \vdash \lambda x \rightarrow e ::_{\downarrow} \tau \rightarrow \tau'} \text{ [LAM]}$$

图 3 λ_{\rightarrow} 的类型规则

我们首先来看可推断项 (*inferable term*) [ANN]，我们将带类型注解的项与其类型注解进行比对，然后返回该类型。变量的类型可以在环境中查找 [VAR]。对于应用 [APP]，我们首先处理函数 e ——它必须具有函数类型，然后我们将参数 e' 与函数的输入类型进行比较，并将函数的返回类型⁴作为结果类型返回。

最后两条规则用于类型检查。如果我们能推断出一个项的类型，且该类型与给定的类型一致，那么该项也能通过给定类型的检查 [CHK]。Lambda 抽象只能被检查为函数类型 [LAM]。我们在扩展后的上下文中检查 Lambda 抽象的函数体 (*body*)。

请注意，这些规则几乎完全是语法制导 (*syntax-directed*) 的：尽管连接可检查项和可推断项的规则 [CHK] 似乎可以匹配任何项，然而规则中并没有用于推断 Lambda 抽象的类型的规则，也没有显式地检查类型注解、变量或应用的规则。因此，这些规则可以被很容易地转换为语法制导的算法。

以下是根据上述规则推出的两个示例 `id` 和 `const` 的类型判断 (*type judgement*)：

$$\alpha :: *, y :: \alpha \quad \vdash (\text{id} :: \alpha \rightarrow \alpha) y :: \alpha$$

$$\alpha :: *, y :: \alpha, \beta :: * \vdash (\text{const} :: (\beta \rightarrow \beta) \rightarrow \alpha \rightarrow \beta \rightarrow \beta) \text{id } y :: \beta \rightarrow \beta$$

⁴译注：“输入类型”原作“定义域 (*domain*)”，“返回类型”原作“值域 (*range*)”。“值域”这一术语并不严谨，既可指陪域/到达域 (*codomain*)，亦可指像/像集 (*image/image set*)。且在类型论语境下，将集合 (*Set*) 与类型 (*Type*) 混为一谈，则容易引出许多的危险。简单起见，译者扬弃了这两个术语，统一使用“输入类型”和“返回类型”。至于相关理论知识，本文限于篇幅不再赘述，感兴趣的读者可自行探索。

2.4. 实现

现在，我们给出一个 λ_{\rightarrow} 的 Haskell 实现。我们提供了一个用于对类型良定的 (*well-typed*) 词项求值的求值器，以及一个用于对 λ_{\rightarrow} 词项进行类型检查的函数。该实现与我们刚才介绍的形式化描述非常吻合。

规则的实现方式有着相当的自由度。我们选择的实现方式既能让我们紧密地遵循类型系统，又能将技术上的心智负担降到最低，让我们能够专注于算法的本质。接下来，我们会简要讨论我们的设计决策，并给出一些替代方案。需要指出的是，对于依值类型的实现来说，任何一种选择都不是必然的。

2.4.1. 变量和值的表示

⁵有很多不同的方法可以被用于表示约束变量 (*bound variable*)，每种方法都各有优劣。为了最大限度地发挥优势，我们在实现的不同地方选择不同的表示方法。

我们使用德布鲁因索引 (*de Bruijn indices*) 表示局部约束变量：变量由数字而非字符串或字母表示，数字的含义是变量出现的位置和引入它的约束符 (*binder*)⁶ 之间隔了多少层约束符。例如，使用德布鲁因索引，`id` 可以写成 $\lambda \rightarrow 0$ ，而 `const` 可以写成 $\lambda \rightarrow \lambda 1$ 。这一表示法的优势在于不需要重命名变量——词项之间的 α -等价性 (α -equivalence)⁷ 可以简化为语法相等性 (*syntactic equality*)。

德布鲁因索引的缺点则是处理带有自由变量的词项相当麻烦。我们可以用未被 Lambda 约束的索引来表示自由变量，但这些索引是相对的 (*relative*)——当我们遍历每个词项、进入 Lambda 表达式内部时，这些索引必须相应地更新。

因此，我们使用绝对指涉——也就是名称 (*name*)——来表示自由变量。这种为局部变量使用数字、为相对于当前词项的全局变量使用变量名的做法称为局部无名 (*locally nameless*) 表示法 [23, 13]。

最后，我们使用高阶抽象语法 (*high-order abstract syntax*) 表示值：使用 Haskell 函数来表示函数值。这样做的优势在于我们可以利用 Haskell 的函数应用，不必自己实现替换，也不用担心名称捕获 (*name capture*)。这种方法的一个小缺点是 Haskell 函数既不能显示，也不能进行相等性 (*equality*) 检查。幸运的是，通过将值引用 (*quote*) 回具体的表示形式，这一缺陷可以被轻易缓解。在我们定义完求值器和类型检查器之后，我们会进一步讨论引用的具体机制。

2.4.2. 分离可推断项和可检查项

正如我们在图 3 的 λ_{\rightarrow} 的类型规则中所揭示的，我们区分了可以读出类型的项（称为可推断项）和需要进行类型检查的项（可检查项 *checkable term*）。可检查项和可推断项之间的这种语法区分至少可以追溯到 Pierce 和 Turner 的研究 [22]。

另一种做法是，我们可以要求抽象语法中的每个 Lambda 绑定变量都必须显式标注类型——这样我们就只拥有可推断项了。然而，能够为任意词项标注类型是非常有用的。既然有了通用的标注机制，就不再需要强制要求每个 Lambda 绑定变量都带类型注解了。事实上，允许未标注的 Lambda 在不增加额外成本的情况下提供了很大的便利：形如 $e(\lambda x \rightarrow e')$ 的函数应用可以在没有类型标注的情况下被处理，因为 x 的类型是由 e 的类型决定的。

⁵译注：本小节标题原为“Representing bound variables”，直译为“约束变量的表示”。然实则本小节讨论内容不仅限于约束变量的表示，故改。

⁶译注：在这里，约束符指的就是 λ 。第三节会引入另一个约束符 \forall/Π 。

⁷译注：“ α -equivalence”原作 α -“equality”。

2.4.3. 抽象语法

我们为可推断项 ($\text{Term}\uparrow$)、可检查项 ($\text{Term}\downarrow$) 和名称 (Name) 引入以下数据类型:

```
data Term↑
  = Ann    Term↓ Type
  | Bound  Int
  | Free   Name
  | App    Term↑ Term↓
  deriving (Show, Eq)

data Term↓
  = Inf Term↑
  | Lam Term↓
  deriving (Show, Eq)

data Name
  = Global String
  | Local  Int
  | Quote  Int
  deriving (Show, Eq)
```

带类型注解的项以 **Ann** 表示。如前所述, 我们用整数表示约束变量 (**Bound**), 用名称表示自由变量 (**Free**)。那些通常指涉全局实体的名称使用字符串 (**Global**)。当类型检查算法⁸进入一层约束符⁹时, 我们需要将约束符引入的约束变量临时转换成一个自由变量, 我们用 **Local** 表示这类变量。进行引用时, 我们使用 **Quote** 构造子。构造子 **App**¹⁰ 表示应用。

可推断项通过构造子 **Inf** 嵌入到可检查项中。Lambda 抽象 (因为我们用了德布鲁因索引, 所以不会引入显式的变量) 使用 **Lam** 表示。

类型只有两种: 类型标识符 (**TFree**) 和函数箭头 (**Fun**)。我们为类型标识符复用 **Name** 数据类型。在 λ_{\rightarrow} 中, 类型层面上不会有约束变量, 所以不需要 **TBound** 构造子。

```
data Type
  = TFree Name
  | Fun   Type Type
  deriving (Show, Eq)
```

值包括 Lambda 抽象 (**VLam**) 和中性项 (**VNeutral**)。

```
data Value
  = VLam      (Value -> Value)
  | VNeutral  Neutral
```

正如我们在讨论高阶抽象语法时所描述的, 我们使用类型为 (**Value** \rightarrow **Value**) 的 Haskell 的函数表示函数值。例如, 项 **const** 求值后就会得到值 **VLam** ($\backslash x \rightarrow \text{VLam } (\backslash y \rightarrow x)$)。

表示中性项的数据类型和形式化的抽象语法完全吻合。一个中性项要么是一个自由变量 (**NFree**), 要么是中性项对值的应用 (**NApp**)。

⁸译注: 原文无此“类型检查”, 依后文补。

⁹译注: 原文为“passing a binder”。因为 **pass** 既可以表示“通过”又可以表示“传递”, 同时又少了递归的意思, 因此此处意译为“进入”。

¹⁰译注: 原文用符号是中缀运算符“ $:@:$ ”, 译者依个人好恶改。

```
data Neutral
  = NFree  Name
  | NApp   Neutral Value
```

我们引入一个函数 `vfree`，用于创建对应于自由变量的值：

```
vfree :: Name -> Value
vfree n = VNeutral (NFree n)
```

2.4.4. 求值

用于求值的代码在图 4 中给出。函数 `eval↑` 和 `eval↓` 实现了可推断项和可检查项的大步求值规则。将代码与图 1 对比，足见这一实现直截了当。

替换是通过传递一个包含值的环境 `Env` 来实现的。因为约束变量是用数字表示的，`Env` 可以被简单地实现成一个列表 `[Value]`，其中第 i 个元素对应着变量 `Bound i`。当进入一层约束符时，我们向环境中（列表的头部）添加一个新元素；当遇到变量 `Bound` 时，我们使用 Haskell 的 `!!` 运算符从列表中拿出正确的元素。

而对于 Lambda 抽象 (`Lam`)，我们引入一个 Haskell 函数，该函数首先将约束变量 x 添加到环境中，然后对函数体求值。

```
type Env = [Value]

eval↑ :: Term↑ -> Env -> Value
eval↑ (Ann e _) d = eval↓ e d
eval↑ (Free x _) d = vfree x
eval↑ (Bound i _) d = d !! i
eval↑ (App e e') d = vapp (eval↑ e d) (eval↓ e' d)

vapp :: Value -> Value -> Value
vapp (VLam f _) v = f v
vapp (VNeutral n) v = VNeutral (NApp n v)

eval↓ :: Term↓ -> Env -> Value
eval↓ (Inf i) d = eval↑ i d
eval↓ (Lam e) d = VLam (\x -> eval↓ e (x : d))
```

图 4 λ_{\rightarrow} 求值器的实现

2.4.5. 上下文

在着手实现类型检查之前，我们定义类型上下文。上下文以（反向的）列表形式实现，将名称与 `*` (`HasKind Star`) 或者一个类型 (`HasType τ`) 关联起来。上下文的扩展自然是通过列表的 `cons` 操作 (`:`) 实现的；在上下文中查找名称则是使用 Haskell 的标准列表函数 `lookup` 进行的。

```
data Kind = Star
  deriving (Show)

data Info
  = HasKind Kind
  | HasType Type
  deriving (Show)

type Context = [(Name, Info)]
```

2.4.6. 类型检查

现在我们实现图 3 中所描述规则，代码如图 5 所示。类型检查算法可能失败，而为了优雅地处理错误，该算法使用 `Result` 单子 (*monad*) 返回结果。简单起见，在本演示中，我们使用标准的错误单子：

```
type Result α = Either String α
```

并使用函数 `throwError :: String -> Result α` 报告错误。

用于可推断项的函数 `type↑` 返回一个类型，而用于可检查项的函数 `type↓` 接受一个类型作为输入，并返回 `()`。类型的良构性由函数 `kind↓` 检验。函数定义中的每个分支都直接对应一条规则。

```
kind↓ :: Context -> Type -> Kind -> Result ()
kind↓ Γ (TFree x) Star
  = case lookup x Γ of
      Just (HasKind Star) -> pure ()
      Nothing             -> throwError "unknown Identifier"
kind↓ Γ (Fun κ κ') Star
  = do kind↓ Γ κ Star
       kind↓ Γ κ' Star

type↑0 :: Context -> Term↑ -> Result Type
type↑0 = type↑ 0

type↑ :: Int -> Context -> Term↑ -> Result Type
type↑ i Γ (Ann e τ)
  = do kind↓ Γ τ Star
       type↓ i Γ e τ
       pure τ
type↑ i Γ (Free x)
  = case lookup x Γ of
      Just (HasType τ) -> pure τ
      Nothing           -> throwError "unknown identifier"
type↑ i Γ (App e e')
  = do σ <- type↑ i Γ e
       case σ of
         Fun τ τ' -> do type↓ i Γ e' τ
                        pure τ'
         _         -> throwError "illegal application"

type↓ :: Int -> Context -> Term↓ -> Type -> Result ()
type↓ i Γ (Inf e) τ
  = do τ' <- type↑ i Γ e
       unless (τ == τ') (throwError "type mismatch")
type↓ i Γ (Lam e) (Fun τ τ')
  = type↓ (i + 1) ((Local i, HasType τ) : Γ)
              (subst↓ 0 (Free (Local i)) e) τ'
type↓ i Γ _ _
  = throwError "type mismatch"
```

图 5 λ_{\rightarrow} 类型检查器的实现¹¹

¹¹译注：pure 原作 return，依现代 Haskell 实践改。

类型检查函数接受一个整数，该整数表示我们进入了多少层约束符。初次调用时，这个参数应该为 0，我们为此提供了包装函数 `type↑0`。我们用这个整数模拟处理约束变量时的类型规则：在 Lambda 抽象的类型规则 [LAM] 中，在检查函数体时，我们将约束变量添加到上下文中——而我们的实现正是这么做的。计数器 `i` 表示我们进入了多少层约束符，因此 `Local i` 总是一个可用的新名称。我们先将 `Local i` 添加到上下文 Γ 中，将其与约束变量关联起来，然后对函数体作类型检查。因为我们把一个约束变量变成了一个自由变量，所以我们要对函数体进行相应的替换。类型检查器不会遇到约束变量，因此函数 `type↑` 没有用于处理 `Bound` 的分支。

请注意，在检查可推断项时，类型的等价性检查是通过对数据类型 `Type` 进行简单的语法相等性判断来实现的。我们的类型检查器不执行合一 (*unification*)。

用于替换的代码如图 6 所示，替换算法同样包含两个函数：用于可推断项的函数 `subst↑` 和用于可检查项的函数 `subst↓`。整数参数表示要替换哪个变量。在 `Bound` 的分支中，我们要检查遇到的变量是不是需要替换的变量。在 `Lam` 的分支中，我们要增加 `i` 的值，因为在 Lambda 抽象的函数体中，待替换变量是由更高的编号指涉的。

```
subst↑ :: Int -> Term↑ -> Term↑ -> Term↑
subst↑ i r (Ann e τ) = Ann (subst↓ i r e) τ
subst↑ i r (Bound j) = if i == j then r else Bound j
subst↑ i r (Free y) = Free y
subst↑ i r (App e e') = App (subst↑ i r e) (subst↓ i r e')

subst↓ :: Int -> Term↑ -> Term↓ -> Term↓
subst↓ i r (Inf e) = Inf (subst↑ i r e)
subst↓ i r (Lam e) = Lam (subst↓ (i + 1) r e)
```

图 6 λ_{\downarrow} 替换算法的实现

2.4.7. 引用

我们的简单类型 Lambda 演算求值器就快完成了。目前还剩一个小问题：求值器返回的是 `Value`，而当下我们无法打印 `Value` 类型的值，也不能对 `Value` 作相等性判断。这是因为 `Value` 类型的 `VLam` 构造子接受的是一个 Haskell 函数，我们不能像对其他类型那样简单地为其派生 `Show` 和 `Eq`¹²。因此，如果要重新得到一个值的内部结构，就需要用到 `quote` 函数。代码在图 7 中给出。

```
quote0 :: Value -> Term↓
quote0 = quote 0

quote :: Int -> Value -> Term↓
quote i (VLam f) = Lam (quote (i + 1) (f (vfree (Quote i))))
quote i (VNeutral n) = Inf (neutralQuote i n)

neutralQuote :: Int -> Neutral -> Term↑
neutralQuote i (NFree x) = boundfree i x
neutralQuote i (NApp n v) = App (neutralQuote i n) (quote i v)
```

图 7 λ_{\downarrow} 中的引用

函数 `quote` 接受一个整数参数，该参数用来记录我们已经进入了多少层约束符。初次调用时，这个参数总是为 0，我们为此提供了包装函数 `quote0`。

¹²译注：此处译文结构较原文有较大变化。

如果传给 `quote` 的值是一个 Lambda 抽象 (`VLam`)，我们生成一个新变量 `Quote i`，把它传给 Haskell 函数 `f`，然后对函数 `f` 返回的值递归调用 `quote (i + 1)`¹³。这里，我们使用接受一个 `Int` 的构造子 `Quote` 来确保新创建的名称不会与值中的其他名称冲突。

如果值是一个中性项 (`VNeutral`，也就是将自由变量应用于零个或多个¹⁴其他值)，则使用 `neutralQuote` 函数处理¹⁵。`boundfree` 函数被用于检查出现在中性项¹⁶头部的变量是一个自由变量，还是一个 `Quote` ——也就是约束变量：

```
boundfree :: Int -> Name -> Term↑
boundfree i (Quote k) = Bound (i - k - 1)
boundfree i x         = Free x
```

用例子来理解函数是如何被引用的再好不过了。与词项 `const` 对应的值是 `VLam (\x -> VLam (\y -> x))`。对其应用 `quote0` 可得：

```
quote 0 (VLam (\x -> VLam (\y -> x)))
= Lam (quote 1 (VLam (\y -> vfree (Quote 0))))
= Lam (Lam (quote 2 (vfree (Quote 0))))
= Lam (Lam (neutralQuote 2 (NFree (Quote 0))))
= Lam (Lam (Bound 1))
```

当 `quote` 进入一层约束符时，我们为约束变量引入一个临时的名称。为确保在引用过程中该名称不会和其他名称发生冲突，我们只使用 `Quote` 构造子。如果约束变量在函数体中出现过，那么我们迟早会抵达这些出现的地方。此时即可根据引入和观测到 `Quote` 构造子之间经过的约束符数量，生成德布鲁因索引。

2.4.8. 例子

略。请自行参照原文。

¹³译注：原文为“The value resulting from the function application is then quoted at level $i + 1$.”译者不想翻译这个“level”，因为这个词的这种用法只出现了一次，且在后文中还有别的意思。同时，尽管译者在翻译所有“refer”的地方都用了“指涉”，“被引用”听起来仍然很别扭。故利用 Haskell 的柯里化特性将其意译。

¹⁴译注：原文无此“零个或多个”，依前文补。

¹⁵译注：原文为“If the value is a neutral term (hence an application of a free variable to other values), the function `neutralQuote` is used to quote the arguments.”此处原文显然逻辑不通（我们对中性项的参数调用的是 `quote` 而非 `neutralQuote`），故译者完全重写了这一句。

¹⁶译注：“中性项”原作“应用”。

3. 依值类型

在本节中，我们将修改简单类型 Lambda 演算的类型系统，使其成为依值类型 Lambda 演算，简称 λ_{Π} 。在本节的开头，我们会先讨论这些更改中的两个核心思想。接着，我们会给出抽象语法、求值和类型规则，并将这些规则中与 λ_{\rightarrow} 不同的部分用高亮标出。最后，我们在本节中讨论如何调整实现。

3.0.1. 依值函数空间

在 Haskell 这样的语言中，我们可以定义多态函数，例如恒等函数：

```
id :: ∀α. α -> α
id = \x -> x
```

利用多态，我们可以避免为不同类型的数据——例如整数和布尔值——反复编写同样的函数。若将多态解释为一种类型抽象，则可使其变成一种显式 (*explicit*) 行为。这样一来，恒等函数就接受两个参数：一个类型 α ，和一个类型为 α 的值。而在调用这个恒等函数时，必须显式地用一个类型将其实例化 (*instantiate*)：

```
id :: ∀α. α -> α
id = \(α :: *) (x :: α) -> x

id Bool True :: Bool
id Int 3     :: Int
```

因此，多态允许类型对类型进行抽象。为什么我们还要做与此不同的事情呢？考虑以下数据类型：

```
data Vec0 α = Vec0
data Vec1 α = Vec1 α
data Vec2 α = Vec2 α α
data Vec3 α = Vec3 α α α
```

显然，这些类型遵循着某种模式。我们希望能有一个单一的类型族 (*family*)，并按元素数量进行索引：

$$\forall \alpha :: *. \forall n :: \text{Nat}. \text{Vec } \alpha \ n$$

但在 Haskell 里我们不能简单地这么做。问题在于，这个类型 **Vec** 是对值 n 抽象的。

依值函数空间 “ \forall ” 扩展了通常的函数空间 “ \rightarrow ”，它允许函数返回值的类型依赖于输入值¹⁷。Haskell 中的参数化多态 (*parametric polymorphism*) 可以看作是依值函数的一种特例¹⁸，这也是我们使用符号 “ \forall ” 的动机¹⁹。但与参数化多态不同的是，依值函数空间不止能对类型进行抽象。上面的 **Vec** 类型是一个有效的依值类型。

值得注意的是，依值函数空间是通常函数空间的泛化。例如，我们可以为应用于上述 **Vec** 类型的恒等函数 `id` 添加这样的类型注解：

$$\forall \alpha :: *. \forall n :: \text{Nat}. \forall v :: \text{Vec } \alpha \ n. \text{Vec } \alpha \ n$$

¹⁷译注：原文为 “The dependent function space ‘ \forall ’ generalizes the usual function space ‘ \rightarrow ’ by allowing the range to depend on the domain”，译者依个人理解进行了意译。例如，依值类型的函数 `zeros :: ∀α :: *. ∀n :: Nat. Vec α n` 返回的类型 `Vec α n` 显然依赖于参数 n 的值。

¹⁸译注：因为类型也可以被视作一种值。下一小节“一切皆项”将详细展开。

¹⁹类型学家称依值函数类型为 Π -类型，并且会这么写： $\Pi \alpha :: *. \Pi n :: \text{Nat}. \text{Vec } \alpha \ n$ 。这也是为什么我们将依值类型 Lambda 演算称为 λ_{Π} 。

注意类型 v 并没有在返回类型中出现：而这就是非依值函数空间，这对 Haskell 程序员来说已经很熟悉了。与其在这种地方引入像 v 这样不必要的变量，我们不妨为非依值的情况使用常规的函数箭头。因此上面的类型注解也可以写成：

$$\forall \alpha :: *. \forall n :: \text{Nat}. \text{Vec } \alpha n \rightarrow \text{Vec } \alpha n$$

在 Haskell 中，程序员可以一定程度上“模拟”出依值类型空间 [11]，例如在类型层面上定义自然数（也就是定义 `Zero` 和 `Succ` 这样的数据类型）。然而，类型层面和价值层面的自然数之间，终究是云泥之异，难越鸿沟，程序员不得不在两个层面之间重复大量的概念。尽管利用高级的类型类编程技巧可以将值层面的东西提升到类型层面，但要用这些类型作计算需要许多额外的努力。而使用依值类型，我们可以直接用值将类型参数化，并且仍然使用常规的求值规则——我们很快会看到的。

3.0.2. 一切皆词项

允许值在类型中自由出现打破了词项、类型和种类 (*kind*) 的分界。不同的层级之间不再有语法上的差异，因为一切皆词项。在 Haskell 中，符号 “`::`” 将实体与不同的语法层级关联起来：在 `0 :: Nat` 中，`0` 在语法上是一个值，而 `Nat` 是一个类型；在 `Nat :: *` 中，`Nat` 在语法上是一个类型，而 `*` 是一个种类。而现在，`*`、`Nat` 和 `0` 都是词项。`0 :: Nat` 和 `Nat :: *` 依然成立，但符号 “`::`” 现在是将词项与词项关联起来。我们还是称 $\rho :: *$ 中的 ρ 为类型，也仍然会称呼 $*$ 为种类，但所有这些实体现在都处于同一个语法层级了。这样一来，所有语言结构就在任何地方都可用了。特别地，我们现在可以在类型和种类的层面进行抽象和应用了。

现在我们已经熟悉了依值类型系统的核心概念。接下来，我们将讨论实现 λ_{Π} 要对 λ_{\rightarrow} 作什么样的修改。

3.1. 抽象语法

我们不再需要区分词项、类型和种类了，所有层级中的所有结构现在都被整合到了词项语言中：

$e, \rho, \kappa ::= e :: \rho$	带类型注解的词项
$*$	类型之类型
$\forall x :: \rho. \rho'$	依值函数空间
x	变量
$e e'$	应用
$\lambda x \rightarrow e$	Lambda 抽象

抽象语法中与 2.1 节有所变动之处以高亮显示。

现在我们也使用符号 ρ 和 κ 指涉作为类型和种类的词项。

原本位于类型和种类的语法规则中的构造现在也被导入了进来。种类 $*$ 现在是一个词项。依值函数空间涵盖了原本的箭头种类和箭头类型。类型变量和词项变量现在也已合二为一。

3.2. 求值

新增的求值规则如图 8 所示。除了与新添加的两个构造相关的规则外，所有规则都和 λ_{\rightarrow} 别无二致。你也许会感到惊讶，求值竟然扩展到了类型！但这正是我们想要的：依值类型的威力正源于将值和类型混合的能力，这样一来我们就能在类型层级上定义函数、进行计算。

$$\frac{}{* \Downarrow *} \quad \text{对 } * \text{ 求值会得到它自身}$$

$$\frac{\rho \Downarrow \tau \quad \rho' \Downarrow \tau'}{\forall x :: \rho. \rho' \Downarrow \forall x :: \tau. \tau'} \quad \text{递归求值依值函数空间}$$

图 8 λ_{Π} 新增的求值规则

相对而言，我们新加入的构造在计算过程中就不那么有趣了：对 $*$ 求值会得到它自身；在依值函数空间中，我们递归地对输入类型和返回类型求值。我们必须相应地扩展值的抽象语法：

$$\begin{array}{l} v, \tau ::= n \quad \text{中性项} \\ | * \quad \text{类型之类型} \\ | \forall x :: \tau. \tau' \quad \text{依值函数空间} \\ | \lambda x \rightarrow v \quad \text{Lambda 抽象} \end{array}$$

现在我们使用符号 τ 指涉作为类型的值。

3.3. 类型系统

在接触类型规则之前，我们得先回顾一下上下文。因为现在一切皆词项，上下文的抽象语法和有效性规则（图 9，与图 2 对比）自然也相应地简化了。

$$\begin{array}{l} \Gamma ::= \varepsilon \quad \text{空上下文} \\ | \Gamma, x :: \tau \quad \text{添加一个变量} \end{array}$$

$$\frac{}{\text{valid}(\varepsilon)} \quad \frac{\text{valid}(\Gamma) \quad \Gamma \vdash \tau :: \Downarrow *}{\text{valid}(\Gamma, x :: \tau)}$$

图 9 λ_{Π} 的类型上下文及其有效性

上下文中现在只有一种形式的条目，也就是说我们总是假设变量有其类型。注意我们在上下文中存储的都是求值后的类型。非空上下文有效性规则中的前提条件 $\Gamma \vdash \tau :: \Downarrow *$ 指涉的不再是一个特殊的类型良构性的判断，而是我们即将定义的类型规则——我们不再需要为类型设置专门的良构性规则了。这一前提条件尤其确保了 τ 中不含有未知的变量。

类型规则在图 10 中给出。类型规则现在与上下文、词项和值相关——所有类型都被尽可能早地求值。和之前一样，我们高亮了规则中和图 3 不同的部分。我们仍然为推断和检查使用不同的符号：当一个类型是输出时，我们用 $::\uparrow$ ；当一个类型是输入时，我们用 $::\downarrow$ 。新的构造 $*$ 和 \forall 属于我们能推断其类型的那一类。和之前讨论 λ_{\rightarrow} 的时候一样，我们假定类型规则中出现的所有上下文都是有效的。

$$\begin{array}{c}
\frac{\Gamma \vdash \rho ::_{\downarrow} * \quad \rho \Downarrow \tau \quad \Gamma \vdash e ::_{\downarrow} \tau}{\Gamma \vdash (e :: \rho) ::_{\uparrow} \tau} \text{ [ANN]} \\
\\
\frac{}{\Gamma \vdash * ::_{\uparrow} *} \text{ [STAR]} \quad \frac{\Gamma \vdash \rho ::_{\downarrow} * \quad \rho \Downarrow \tau \quad \Gamma, x :: \tau \vdash \rho' ::_{\downarrow} *}{\Gamma \vdash \forall x :: \rho. \rho' ::_{\uparrow} *} \text{ [PI]} \\
\\
\frac{\Gamma(x) = \tau}{\Gamma \vdash x ::_{\downarrow} \tau} \text{ [VAR]} \quad \frac{\Gamma \vdash e ::_{\uparrow} \forall x :: \tau. \tau' \quad \Gamma \vdash e' ::_{\downarrow} \tau \quad \tau'[x \mapsto e'] \Downarrow \tau''}{\Gamma \vdash e e' ::_{\downarrow} \tau''} \text{ [APP]} \\
\\
\frac{\Gamma \vdash e ::_{\uparrow} \tau}{\Gamma \vdash e ::_{\downarrow} \tau} \text{ [CHK]} \quad \frac{\Gamma, x :: \tau \vdash e ::_{\downarrow} \tau'}{\Gamma \vdash \lambda x \rightarrow e ::_{\downarrow} \forall x :: \tau. \tau'} \text{ [LAM]}
\end{array}$$

图 10 λ_{Π} 的类型规则

带类型注解的词项的规则 [ANN] 有两项变动：注解 ρ 的种类检查规则不再指涉到类型良构性规则，而是指涉到类型检查规则；同时因为注解 ρ 现在未必是一个值，我们需要先对它求值，然后返回求值结果。

种类 $*$ 自身的类型就是 $*$ [STAR]。尽管这种选择存在一些理论上的反对意见（见第五节），我们相信就本文而言，相对于实现的简单性，这些反对意见无足轻重。

依值函数空间的规则 [PI] 和图 2 中箭头类型的良构性规则 [FUN] 有相似之处。依值函数的输入类型 ρ 和返回类型 ρ' 都必须归属于种类 $*$ 。与之前的规则 [FUN] 有所不同的是， ρ' 中可以包含²⁰ x ，因此我们在检查 ρ' 的时候要将 $x :: \tau$ （其中 τ 是对 ρ ²¹ 求值的结果）加入上下文 Γ 。

在函数应用 [APP] 中，现在函数必须具有依值函数类型 $\forall x :: \tau. \tau'$ 。与普通函数类型不同的是， τ' 中可以包含 x 。因此在应用的结果类型中，我们需要将 τ' 中出现的形参 x 替换成实参 e' 。

对可推断项作检查的规则 [CHK] 还是跟以前一样：给定一个词项 e 和一个类型 τ ，我们首先推断出 e 的类型，接着检查推断出的类型 and 期望的类型 τ 是否相等。然而，我们现在处理的是已求值的类型，因此这种相等性要比类型词项的语法相等性强得多：不然要是 $\text{Vec } \alpha 2$ 和 $\text{Vec } \alpha (1 + 1)$ 表示的不是同一个类型，那就太不幸了。而我们的系统能识别出它们相等，因为两个类型的求值结果都是 $\text{Vec } \alpha 2$ 。

许多支持依值类型的类型系统都有一条这样的规则：

$$\frac{\Gamma \vdash e :: \rho \quad \rho =_{\beta} \rho'}{\Gamma \vdash e :: \rho'}$$

然而，这条被称作转换规则 (*conversion rule*) 的规则显然是非语法制导的，而区分可推断项和可检查项使得我们只在处理有显式类型注解的词项 [ANN] 时才需要应用这条转换规则。

最后一条规则 [LAM] 是用来检查 Lambda 抽象的。与之前不同的是，Lambda 抽象的类型现在是一个依值函数类型，约束变量 x 不仅可能出现在函数体 e 中，还可能出现在返回类型 τ' 中²²。因此在对函数体 e 作类型检查和对返回类型 τ' 作种类检查时，都要使用扩展过的上下文 $\Gamma, x :: \tau$ 。

总结下来，我们所作的所有修改都是围绕着我们第三节开头引入的两个核心概念进行的：函数空间被泛化为依值函数空间；类型和种类也是词项。

²⁰译注：“包含”原作“指涉”。

²¹译注：“ ρ ”原作“ τ' ”，但规则 [PI] 中并无 τ' ，故依实际情况改写为 ρ 。

²²译注：后半句“还可能出现在返回类型 τ' 中”为译者补文。

3.4. 实现

我们给出的 λ_{Π} 的类型规则仍然是语法制导的、算法性的，所以 λ_{\rightarrow} 实现的总体结构可以被 λ_{Π} 复用。在接下来的部分里，我们会遍历实现的各个方面，但只讨论需要修改的部分。

3.4.1. 抽象语法

现在我们不再需要 **Type** 和 **Kind** 了。我们为 **Term** 添加了两个构造子，并将构造子 **Ann** 中出现的 **Type** 替换成了 **Term \downarrow** ：

```
data Term $\uparrow$ 
  = Ann    Term $\downarrow$  Term $\downarrow$ 
  | Star
  | Pi     Term $\downarrow$  Term $\downarrow$ 
  | Bound Int
  | Free  Name
  | App   Term $\uparrow$  Term $\downarrow$ 
```

我们也需要扩展用来表示值的类型：

```
data Value
  = VLam    (Value  $\rightarrow$  Value)
  | VStar
  | VPi     Value (Value  $\rightarrow$  Value)
  | VNeutral Neutral
```

和之前一样，我们用高阶抽象语法表示值，也就是用 Haskell 函数来表示约束结构²³。我们用 **VPi** 表示新的约束结构 \forall/Π 。在依值函数空间中，由 $\forall x : A. B$ 引入的约束变量 x 不会出现在变量自身的类型 A 中，但函数的返回类型 B 中却有可能包含 x ²⁴。因此，输入类型可以简单地用一个 **Value** 表示，而返回类型则要用 Haskell 函数 **Value \rightarrow Value** 表示²⁵。

3.4.2. 求值

要适配求值器，我们只需为函数 **eval \uparrow** 添加两个用于处理 **Star** 和 **Pi** 的新分支，如图 11 所示（ λ_{\rightarrow} 的求值器见图 4）。**Star** 的求值非常简单。对于 **Pi**，我们求值其输入类型和返回类型，且在求值返回类型时，我们需要将约束变量 x 添加到上下文中。

```
eval $\uparrow$  Star      d = VStar
eval $\uparrow$  (Pi  $\tau$   $\tau'$ ) d = VPi (eval $\downarrow$   $\tau$  d) (\x  $\rightarrow$  eval $\downarrow$   $\tau'$  (x : d))

subst $\uparrow$  i r (Ann e $\downarrow$   $\tau$ ) = Ann (subst $\downarrow$  i r e $\downarrow$ ) (subst $\downarrow$  i r  $\tau$ )
subst $\uparrow$  i r Star        = Star
subst $\uparrow$  i r (Pi  $\tau$   $\tau'$ ) = Pi (subst $\downarrow$  i r  $\tau$ ) (subst $\downarrow$  (i + 1) r  $\tau'$ )

quote i VStar      = Inf Star
quote i (VPi v f) = Inf (Pi (quote i v)
                           (quote (i + 1) (f (vfree (Quote i)))))
```

图 11 λ_{Π} 对求值、替换和引用的扩展

²³译注：也就是 λ 和 \forall/Π 。从现在开始， \forall 也是约束符之一了。

²⁴译注：此句为译者改写。原文为 “In the dependent function space, a variable is bound that is visible in the range, but not in the domain.”

²⁵译注：不妨把 $\Pi x : A. B$ 看作 $\Pi x : A. B(x)$ ，即类型 B 是关于值 x 的函数。

3.4.3. 上下文

上下文将变量映射到其类型，而类型现在也在词项层级上。我们存储的是类型求值后的形式，因此我们如是定义类型和上下文：

```
type Type = Value
type Context = [(Name, Type)]
```

3.4.4. 类型检查

现在我们逐一分析图 12 中的分支，你可以将其与图 5 对比。

```
type↑ :: Int -> Context -> Term↑ -> Result Type
type↑ i Γ (Ann e ρ)
  = do type↓ i Γ ρ VStar
      let τ = eval↓ ρ []
      type↓ i Γ e τ
      pure τ
type↑ i Γ Star
  = pure VStar
type↑ i Γ (Pi ρ ρ')
  = do type↓ i Γ ρ VStar
      let τ = eval↓ ρ []
      type↓ (i + 1) ((Local i, τ) : Γ)
                (subst↓ 0 (Free (Local i)) ρ') VStar
      pure VStar
type↑ i Γ (Free x)
  = case lookup x Γ of
      Just τ -> pure τ
      Nothing -> throwError "unknown identifier"
type↑ i Γ (App e e')
  = do σ <- type↑ i Γ e
      case σ of
        VPi τ τ' -> do type↓ i Γ e' τ
                        pure (τ' (eval↓ e' []))
        _ -> throwError "illegal application"

type↓ :: Int -> Context -> Term↓ -> Type -> Result ()
type↓ i Γ (Inf e) v
  = do v' <- type↑ i Γ e
      unless (quote0 v == quote0 v') (throwError "type mismatch")
type↓ i Γ (Lam e) (VPi τ τ')
  = type↓ (i + 1) ((Local i, τ) : Γ)
            (subst↓ 0 (Free (Local i)) e) (τ' (vfree (Local i)))
type↓ i Γ _ _
  = throwError "type mismatch"
```

图 12 λ_{Π} 类型检查器的实现

对于注解项 $e :: \rho$ ，我们首先用类型检查函数 `type↓` 检查类型注解 ρ 具有种类 $*$ 。接着我们对 ρ 求值，用求值结果 τ 对 e 作类型检查，如果类型检查成功，整个表达式的类型就是 τ ²⁶。注意我们假设 `type↑` 所处理的类型中没有非约束变量，因此我们总是传给 `eval↓` 一个空环境。

`Star` 求值后的类型是 `VStar`。

²⁶译注：“ τ ”原作“ v ”，然此处显然为作者笔误。

对于依值函数类型 $\forall x :: \rho. \rho'$ ，我们首先对输入类型 ρ 作种类检查，然后将其求值为 τ^{27} 。接着，我们将值 τ 加入上下文，对返回类型 ρ' 作种类检查——这里的思路跟 λ_{\rightarrow} 和 λ_{Π} 中对 **Lam** 作类型检查的规则有异曲同工之妙。

对于函数应用 ee' ，类型推断函数 **type \uparrow** 现在会给出一个值。该值的形式必须是 **VPi** $\tau \tau'$ ——也就是依值函数类型。在图 10 相应的类型规则中，返回类型 τ' 中的约束变量要用 e' 替换。而在实现中， τ' 是一个函数，替换是通过将 τ' 应用于（求值后的） e' 实现的。

在处理 **Inf** 时，我们必须对给定的 v 和推断出的 v' 作类型相等性判断。与类型规则不同的是，在 Haskell 中我们不能直接比较两个 **Value**。因此我们要用 **quote** 将这两个值引用回词项，然后再对词项作语法相等性判断。

对于 Lambda 抽象 $\lambda x \rightarrow e$ ，现在我们要要求其具有依值函数类型 **VPi** $\tau \tau'$ 。和 λ_{\rightarrow} 一样，我们要在对函数体 e 作类型检查时将约束变量 x （类型为 τ ）加入上下文；但与 λ_{\rightarrow} 不同的是，现在除了要用 **subst \downarrow** 替换函数体 e 中的 x 之外，还要将 τ' 应用于 (**Local** i) 来替换 τ' 中的 x 。

为此，我们还要扩展替换函数，使其能够遍历注解项中的类型，并能够处理新结构 **Star** 和 **Pi**，如图 11 所示。对于 **Star**，没有需要替换的东西。而对于 **Pi**，我们要在对返回类型作替换时增加计数器的值，因为我们进入了一层约束符。

3.4.5. 引用

现在只要再扩展引用函数 **quote**，我们的 λ_{Π} 实现就大功告成了。引用操作在 λ_{Π} 中比在 λ_{\rightarrow} 中更加重要，因为如我们所见，现在类型检查的过程中要作相等性检查，而相等性检查需要用到 **quote**。我们还是只需要处理新结构 **VStar** 和 **VPi**，如图 11 所示。

引用 **VStar** 会得到 **Star**。而因为依值函数类型是一个约束结构，引用 **VPi** $\tau \tau'$ 的过程就类似于引用 **VLam**：在引用返回类型 τ' 时，我们将代表着函数返回类型的 Haskell 函数 τ' 应用于 **Quote** i ，然后对结果应用 **quote** ($i + 1$)²⁸。

3.5. 依值类型今何在？

现在我们已经实现了依值类型系统，但不幸的是，我们还没见到任何例子。

和之前一样，我们为 λ_{Π} 的类型检查器写了一个小的解释器，我们可以用它定义实体、执行类型检查。例如，我们可以这样定义和检查多态恒等函数（类型参数是显式的）：

```

>> let id = ( $\lambda \alpha x \rightarrow x$ ) ::  $\forall (\alpha :: *) . \alpha \rightarrow \alpha$ 
id ::  $\forall (x :: *) (y :: x) . x$ 
>> assume (Bool :: *) (False :: Bool)
>> id Bool
 $\lambda x \rightarrow x$  ::  $\forall x :: \text{Bool} . \text{Bool}$ 
>> id Bool False
False :: Bool

```

相比于简单类型，我们能做的事情更多，但其中并没有非用依值类型不可的。不幸地，尽管我们已经有了依值类型的框架，如果我们不为我们的语言添加一些特定的数据类型的话，我们就写不出什么有意思的程序。

²⁷译注：原文为 “For a dependent function type, we first kind-check the domain τ . Then the domain is evaluated to v .” 此处显然为作者笔误。

²⁸译注：原文为 “To quote the range, we increment the counter i , and apply the Haskell function representing the range to **Quote** i .”

4. λ_{Π} 之后

在 Haskell 中，数据类型是通过特殊的 `data` 声明引入的：

```
data Nat = Zero | Succ Nat
```

这一声明引入了新类型 `Nat`，以及它的两个构造子 `Zero` 和 `Succ`。而在本节中，我们将探讨如何用数据类型——例如自然数——来扩展我们的语言。

显然，我们需要添加类型 `Nat` 以及它的两个构造子。但我们要如何定义诸如加法的用来操作数字的函数呢？在 Haskell 中，我们可以让函数对参数作模式匹配，并递归调用自身来处理更小的数：

```
plus :: Nat -> Nat -> Nat
plus Zero      n = n
plus (Succ k) n = Succ (plus k n)
```

然而，我们的演算里既没有模式匹配，函数也不能递归。这下可怎么定义 `plus` 呢？

在 Haskell 中，我们可以用 `fold` [16] 为数据类型定义递归函数。所以，相比于引入会带来一堆问题的模式匹配和递归，不如用 `fold` 来为自然数定义函数。不过在依值类型的世界，我们可以定义一个更通用的 `fold`，我们称之为消去子 (*eliminator*)。

自然数的 `fold` 函数具有如下类型：

$$\text{foldNat} :: \forall \alpha :: *. \alpha \rightarrow (\alpha \rightarrow \alpha) \rightarrow \text{Nat} \rightarrow \alpha$$

这应该很熟悉。不过，在依值类型环境中，类型 α 无需在自然数的各个构造子之间保持一致²⁹，因此我们使用 $m :: \text{Nat} \rightarrow *$ 而非 $\alpha :: *$ 。如此，我们可以给出 `natElim` 的类型³⁰：

$\text{natElim} :: \forall m :: \text{Nat} \rightarrow *$	动机
• $m \text{ Zero}$	基准情况
$\rightarrow (\forall l :: \text{Nat}. m l \rightarrow m (\text{Succ } l))$	归纳情况
$\rightarrow \forall k :: \text{Nat}$	待消数
• $m k$	返回类型

消去子的第一个参数有时也被称为动机 (*motive*) [10]，它描述了我们尽除凡数的缘由³¹。第二个参数对应基准情况 (*base case*)，它的类型是我们把自然数 `Zero` 传给 m 得到的。第三个参数对应归纳情况 (*inductive case*)，它的类型是我们把自然数 `Succ l` 传给 m 得到的。在归纳情况中，我们必须描述如何从自然数 l 和类型 $m l$ 构造出一个类型为 $m (\text{Succ } l)$ 的东西。最后一个参数就是我们要消去的数字。`natElim` 的返回类型就是 $m k$ 。

借助上述讨论中获得的启发，我们可以为自然数给出如图 13 和 14 所示的求值规则和类型规则。注意我们不把 `natElim` 当作一个函数，而是当作一个词项形成符 (*term former*)：它不能像函数那样部分应用，必须参数饱和才能构成有效的词项。求值规则所说明的行为和 `fold` 如出一辙。对自然数 k 求值可能会得到一个中性项，我们需要一条规则来处理这种情况：当遇到这种情况时，消去子的求值会卡住，而求值结果中会包含（无法被 β -规约的）消去子 `natElim` 的应用。

²⁹译注：例如，普通的 `fold` 所接受的函数必须为 `Nat` 的两个构造子 `Zero` 和 `Succ` 返回同一个类型的值；消去子 `natElim` 所接受的函数则不然。

³⁰译注：译者改写了公式，使 `natElim` 的每个输入类型和返回类型都独占一行，并在每个类型旁都加了解释。

³¹译注：就是说，“你想把自然数变成什么别的类型？”

$$\begin{array}{c}
\frac{}{\text{Nat} \Downarrow \text{Nat}} \quad \frac{}{\text{Zero} \Downarrow \text{Zero}} \quad \frac{k \Downarrow l}{\text{Succ } k \Downarrow \text{Succ } l} \\
\frac{k \Downarrow \text{Zero} \quad mz \Downarrow v}{\text{natElim } m \text{ } mz \text{ } ms \text{ } k \Downarrow v} \quad \frac{k \Downarrow \text{Succ } l \quad ms \text{ } l \text{ } (\text{natElim } m \text{ } mz \text{ } ms \text{ } l) \Downarrow v}{\text{natElim } m \text{ } mz \text{ } ms \text{ } k \Downarrow v} \\
\frac{k \Downarrow n}{\text{natElim } m \text{ } mz \text{ } ms \text{ } k \Downarrow \text{natElim } m \text{ } mz \text{ } ms \text{ } n}
\end{array}$$

图 13 自然数的求值

$$\begin{array}{c}
\frac{}{\Gamma \vdash \text{Nat} ::_{\uparrow} *} \quad \frac{}{\Gamma \vdash \text{Zero} ::_{\uparrow} \text{Nat}} \quad \frac{\Gamma \vdash k ::_{\downarrow} \text{Nat}}{\Gamma \vdash \text{Succ } k ::_{\uparrow} \text{Nat}} \\
\Gamma \vdash m ::_{\downarrow} \text{Nat} \rightarrow * \\
m \text{ Zero} \Downarrow \tau \quad \Gamma \vdash mz ::_{\downarrow} \tau \\
\forall l :: \text{Nat} . m \text{ } k \rightarrow m \text{ } (\text{Succ } l) \Downarrow \tau' \quad \Gamma \vdash ms ::_{\downarrow} \tau' \\
\frac{\Gamma \vdash k ::_{\downarrow} \text{Nat}}{\Gamma \vdash \text{natElim } m \text{ } mz \text{ } ms \text{ } k ::_{\uparrow} m \text{ } k}
\end{array}$$

图 14 自然数的类型规则

4.1. 实现自然数

总结一下，向我们的语言中添加自然数需要添加三样东西：类型 `Nat`，构造子 `Zero` 和 `Succ`，以及消去子 `natElim`。我们将扩展抽象语法，并为求值和类型检查函数添加新的分支。这些新的分支不需要修改现有代码，所以我们只关注新代码片段。

4.1.1. 抽象语法

我们如是扩展抽象语法：

```
data Term↑ = ...
| Nat
| Zero
| Succ Term↓
| NatElim Term↓ Term↓ Term↓ Term↓
```

我们向 `Term↑` 添加了四个构造子：数据类型 `Nat`，数据构造子 `Zero` 和 `Succ`，以及消去子 `NatElim`。构造子 `NatElim` 是完全应用的：它不接受更多的参数。

4.1.2. 求值

之前，值只有两种，Lambda 抽象和“卡住”的应用。而现在，我们需要扩展用于表示值的数据类型来适配自然数的构造子：

```
data Value = ...
| VNat
| VZero
| VSucc Value
```

引入消去子会让求值变得复杂。当待消自然数无法求值成构造子时，消去子的求值就会卡住。相应地，我们要为这种情况扩展用于表示中性项的数据类型：

```
data Neutral = ...
| NNatElim Value Value Value Neutral
```

图 15 中求值的实现严格遵循了图 13 中的规则。

```

eval↑ Nat      d = VNat
eval↑ Zero     d = VZero
eval↑ (Succ k) d = VSucc (eval↓ k d)
eval↑ (NatElim m mz ms k) d
  = let mzVal = eval↓ mz d
      msVal = eval↓ ms d
      rec kVal = case kVal of
          VZero      -> mzVal
          VSucc l    -> vapp (vapp msVal l) (rec l)
          VNeutral k -> VNeutral (NNatElim (eval↓ m d) mzVal msVal k)
          -          -> error "internal: eval natElim"
      in rec (eval↓ k d)

```

图 15 为自然数扩展求值器

其中，消去子是唯一值得注意的情况。本质上，消去子会被求值为一个 Haskell 函数 `rec`，其行为合乎预期：若待消数求值为 `VZero`，则求值基本情况 `mz`；若待消数求值为 `VSucc l`，则将 `ms` 应用于前驱 `l` 和对消去子的递归调用 `rec l`；最后，若待消数求值为中性项，则整个 `natElim` 的求值结果亦为中性项。如果待消数既不是自然数也不是中性项，这在类型检查阶段就会导致类型错误。因此，最后的兜底分支永远都不会被执行。

4.1.3. 类型

图 16 包含了用于处理自然数的类型检查器的实现。

```

type↑ i Γ Nat = pure VStar
type↑ i Γ Zero = pure VNat
type↑ i Γ (Succ k) =
  do type↓ i Γ k VNat
  pure VNat
type↑ i Γ (NatElim m mz ms k) =
  do type↓ i Γ m (VPi VNat (const VStar))
  let mVal = eval↓ m []
  type↓ i Γ mz (vapp mVal VZero)
  type↓ i Γ ms (VPi VNat
    (\l -> VPi (vapp mVal l)
      (\_ -> vapp mVal (VSucc l))))
  type↓ i Γ k VNat
  let kVal = eval↓ k []
  pure (vapp mVal kVal)

```

图 16 为自然数扩展类型检查器

对构造子 `Zero` 和 `Succ` 的类型检查过程相当直截了当。但对消去子 `natElim` 的类型检查就有些复杂了。回想一下，消去子的类型是：

<code>natElim :: ∀m :: Nat → *</code>	动机
• <code>m Zero</code>	基准情况
<code>→ (∀l :: Nat . m l → m (Succ l))</code>	归纳情况
<code>→ ∀k :: Nat</code>	待消数
• <code>m k</code>	返回类型

首先我们检查并求值动机 m 。拿到 m 的值之后，我们就可以检查基准情况和归纳情况了。用于处理 `Zero` 的函数 mz 应具有类型 $m \text{ Zero}$ ，而用于处理任意自然数后继的函数 ms 应具有类型 $\forall l :: \text{Nat} . m l \rightarrow m (\text{Succ } l)$ 。抛开手动输入这个复杂类型的细枝末节，思路本身并不复杂³²。最后，我们对 k 作检查，确保它确实是一个自然数。整个 `natElim` 的返回类型就是将动机 m 的值应用于待数 k 的值得到的结果。

4.1.4. 其他函数

要完成自然数的定义，我们必须相应地扩展用于替换和引用的辅助函数。不过这些新代码相当简单直接，因为我们没有引入新的约束结构。

4.1.5. 加法

有了手头的这些东西之后，我们终于可以在我们的解释器里定义加法了：

```

>> let plus = natElim (λ _ → Nat → Nat)
                        (λ n → n)
                        (λ k rec n → Succ (rec n))
plus :: ∀(x :: Nat) (y :: Nat) . Nat

```

我们通过对加法的第一个参数进行消去，来定义函数 `plus`。我们为基准情况和归纳情况定义的函数的类型都是 $\text{Nat} \rightarrow \text{Nat}$ ，我们依此设定动机为 $\lambda_ \rightarrow \text{Nat} \rightarrow \text{Nat}$ 。在基准情况中，我们需要将 0 加到参数 n 上——也就是直接返回 n 。在归纳情况下，传入的参数分别是前驱 k 、递归调用 `rec`（对应于操作 `plus k`）和数字 n ，而我们要把数字 n 和 `Succ k` 相加。我们调用 `rec` 将 n 与 k 相加，然后在结果上再套一层 `Succ`。

在定义了 `plus` 之后，我们就可以在解释器里求值简单的加法了：

```

>> plus 40 2
42 :: Nat

```

4.2. 实现向量

只有自然数还是算不得亦可赛艇：我们在 Haskell 里也能轻松写出这样的数据类型。所以作为真正用到依值类型的例子，我们接下来会展示如何实现向量。

和自然数一样，我们要为向量定义三个组件：向量类型、构造子和消去子。我们之前已经给出过向量的类型，它有一个类型和一个自然数作为参数：

$$\forall \alpha :: *. \forall k :: \text{Nat} . \text{Vec } \alpha k :: *$$

向量的构造子和 Haskell 的列表 `List` 很像，唯一的区别在于向量的构造子会将长度信息记录在构造出的类型中：

```

Nil  :: ∀α :: *. Vec α Zero
Cons :: ∀α :: *. ∀k :: Nat . α → Vec α k → Vec α (Succ k)

```

³²译注：原文为“Despite the apparent complication resulting from having to hand code complex types, type checking these branches is exactly what would happen when type checking a fold over natural numbers in Haskell.”译者依个人好恶进行了改写。

向量的消去子本质上和列表的 `foldr` 相同，但它的类型更加通用³³（因而也更复杂）：

<code>vecElim :: ∀α :: *</code>	元素类型
• <code>∀m :: (∀k :: Nat. Vec α k → *)</code>	动机
• <code>m Zero (Nil α)</code>	基准情况
→ <code>(∀l :: Nat. ∀x :: α. ∀xs :: Vec α l.</code>	
<code>m l xs → m (Succ l) (Cons α l x xs))</code>	归纳情况
→ <code>∀k :: Nat. ∀xs :: Vec α k</code>	待消向量
• <code>m k xs</code>	返回类型

整个消去子的类型是对向量的元素类型 α 泛化³⁴的。紧随其后的参数便是动机，和自然数的消去子 `natElim` 一样，`vecElim` 的动机本质上是一个接受向量 (`Vec α k`)、返回类型（种类 $*$ ）的函数，而因为向量类型有其长度 k 作为参数，所以动机还需要一个 `Nat` 型的参数 ($\forall k :: \text{Nat}$)³⁵。接下来的两个参数对应于 `Vec` 的两个构造子。构造子 `Nil` 表示空向量，因此对应于基准情况的参数的类型就是 `m Zero (Nil α)`。用于处理归纳情况——也就是构造子 `Cons`——的参数，则接受一个数字 l 、一个类型为 α 的元素 x 、一个长度为 l 的向量 xs ，以及递归应用消去子的结果，其类型为 $m l xs$ 。它需要将这些元素结合起来，构造出符合所需类型 `m (Succ l) (Cons α l x xs)` 的一个词项，该类型对应于长度为 `Succ l`、内容为 $x : xs$ 的向量³⁶。在向 `vecElim` 提供了这些参数之后，我们就得到了一个能消去任意长度向量的函数。

这个消去子的类型看起来相当复杂，不过如果我们把它和列表的 `foldr` 函数对比：

$$\text{foldr} :: \forall \alpha :: *. \forall m :: *. m \rightarrow (\alpha \rightarrow m \rightarrow m) \rightarrow [\alpha] \rightarrow m$$

我们可以看到，它们的结构是一样的。`vecElim` 的签名看似盘根错节，其实这种复杂性很大程度上只是因为动机以向量为参数，而向量又以自然数为参数。

事实上，不是所有的参数都是必须的——有些参数能从其他参数中推断出来。这种推断能大幅削减语法噪音，用消去子编写程序也就不再寸步难行。所以 λ_{Π} 看似繁文缛节，实则是设计使然——我们有意地把它设计成了一种显式的、低级的语言。

4.2.1. 抽象语法

和自然数一样，我们要扩展抽象语法。我们向 `Term↑` 中加入向量的类型、构造子和消去子：

```
data Term↑ = ...
  | Vec Term↓ Term↓
  | Nil Term↓
  | Cons Term↓ Term↓ Term↓ Term↓
  | VecElim Term↓ Term↓ Term↓ Term↓ Term↓ Term↓
```

注意构造子 `Nil` 也有一个参数，这是因为两个构造子在元素类型上都是多态的，它们都需要元素类型作为参数。对于向量和许多其他数据类型而言，构造子的归类有一定的灵活性：我们可以略去 `Nil` 和 `Cons` 的类型参数以及 `Cons` 的长度参数，并将构造子作为可检查项 `Term↓` 而不是可推断项 `Term↑`。这样我们就可以在应用这两个构造子时少传些参数，代价则是我们必须显式注明向量表达式的类型。

³³译注：原文为 “but its type is a great deal more specific”，但依前文，更具体的显然是 `foldr`，故改。

³⁴译注：“泛化”原作“量化” (quantified)。

³⁵译注：原文为 “As was the case for natural numbers, the motive is a type (kind $*$) parameterized by a vector. As vectors are themselves parameterized by their length, the motive expects an additional argument of type `Nat`.”

³⁶译注：原文为 “It combines those elements to form the required type, for the vector of length `Succ l` where x has been added to xs .”

我们还需要扩展用于表示值和中性项的数据类型：

```
data Value = ...
  | VNil Value
  | VCons Value Value Value Value
  | VVec Value Value

data Neutral = ...
  | NVecElim Value Value Value Value Value Value Value
```

4.2.2. 求值

Vec 类型的构造器 **VNil** 和 **VCons** 的求值是按结构进行的，即将构造子中的词项求值为值。同样，唯一值得注意的情况是消去子的求值，如图 17 所示。如前所述，其行为类似于列表上的 **fold** 操作：我们根据待消向量是 **VNil** 还是 **VCons** 应用相应的参数。在 **VCons** 的分支中，我们还要对向量的尾部 **xs**（长度为 1）递归地调用消去子。若待消向量是一个中性项，则消去子无法被规约，返回的结果亦是中性项。

```
eval↑ (VecElim α m mn mc k xs) d =
  let mnVal = eval↓ mn d
      mcVal = eval↓ mc d
      rec kVal xsVal =
        case xsVal of
          VNil _      -> mnVal
          VCons _ l x xs -> foldl vapp mcVal [l, x, xs, rec l xs]
          VNeutral n    -> VNeutral (NVecElim (eval↓ α d)
                                                (eval↓ m d)
                                                mnVal
                                                mcVal
                                                kVal
                                                n)
                        -> error "internal: eval vecElim"
  in rec (eval↓ k d) (eval↓ xs d)
```

图 17 为向量实现求值

4.2.3. 类型检查

我们扩展了类型检查器，如图 18 所示。代码相对较长，但只要记住每个结构中每个元素该是什么类型，理解起来还是相对容易的。

和自然数一样，我们省略替换和引用的新代码，因为它们实现起来直截了当。

```

type↑ i Γ (Vec α k) =
  do type↓ i Γ α VStar
  type↓ i Γ k VNat
  pure VStar
type↑ i Γ (Nil α) =
  do type↓ i Γ α VStar
  let αVal = eval↓ α []
  pure (VVec αVal VZero)
type↑ i Γ (Cons α k x xs) =
  do type↓ i Γ α VStar
  let αVal = eval↓ α []
  type↓ i Γ k VNat
  let kVal = eval↓ k []
  type↓ i Γ x αVal
  type↓ i Γ xs (VVec αVal kVal)
  pure (VVec αVal (VSucc kVal))
type↑ i Γ (VecElim α m mn mc k vs) =
  do type↓ i Γ α VStar
  let αVal = eval↓ α []
  type↓ i Γ m
    (VPi VNat (\k -> VPi (VVec αVal k) (\_ -> VStar)))
  let mVal = eval↓ m []
  type↓ i Γ mn
    (foldl vapp mVal [VZero, VNil αVal])
  type↓ i Γ mc
    (VPi VNat (\l ->
      VPi αVal (\y ->
        VPi (VVec αVal l) (\ys ->
          VPi (foldl vapp mVal [l, ys]) (\_ ->
            (foldl vapp mVal [VSucc l, VCons αVal l y ys]))))))))
  type↓ i Γ k VNat
  let kVal = eval↓ k []
  type↓ i Γ vs (VVec αVal kVal)
  let vsVal = eval↓ vs []
  pure (foldl vapp mVal [kVal, vsVal])

```

图 18 为向量扩展类型检查器

4.2.4. append

现在我们可以用实际行动来阐述真正的依值类型程序了：我们要写一个能将两个向量连接起来，并正确记录新向量长度的函数。在解释器中，这个函数可以像这样定义：

```

>> let append =
  (λ α → vecElim α
    (λ m _ → ∀ (n :: Nat). Vec α n → Vec α (plus m n))
    (λ _ v → v)
    (λ m v vs rec n w → Cons α (plus m n) v (rec n w)))
  :: ∀ (α :: *) (m :: Nat) (v :: Vec α m) (n :: Nat) (w :: Vec α n).
    Vec α (plus m n)

```

和 `plus` 一样，我们通过消去二元函数的第一个参数，来为向量定义二元函数 `append`，因此动机要接受第二个向量。`append` 返回向量的长度应是两输入向量长度之和 `plus m n`。向空向量附加另一向量 v^{37} 的结果是 v 。向形如 `Cons m v vs` 的向量附加向量 w^{38} 则须先调用 `rec` 递归处理 vs ——它会向 vs 附加 w ，然后再在 `rec` 结果的前面附上 v^{39} 。

在定义了函数之后，我们就可以使用它了：

```

>> assume (α :: *) (x :: α) (y :: α)
>> append α 2 (Cons α 1 x (Cons α 0 x (Nil α)))
  1 (Cons α 0 y (Nil α))
Cons α 2 x (Cons α 1 x (Cons α 0 y (Nil α))) :: Vec α 3

```

我们先假设存在类型 α 以及类型为 α 的两个元素 x 和 y ，然后向包含两个 x 的向量附加包含一个 y 的向量。

4.3. 讨论

在本节中我们向核心演算中添加了两个数据类型：自然数和向量。运用同样的方法，我们可以添加许多其他的数据类型。例如，对于任意自然数 n ，我们可以定义一个类型 `Fin n`，该类型包含且仅包含 n 个元素。特别地，`Fin 0`、`Fin 1`、`Fin 2` 分别就是空类型、单位 (*unit*) 类型和布尔类型。此外，`Fin` 还可以被用来定义一个全函数 (*total function*)⁴⁰ 版本的投影 (*projection*)⁴¹ 函数：

$$\text{project} :: \forall (\alpha :: *) (n :: \text{Nat}). \text{Vec } \alpha \ n \rightarrow \text{Fin } n \rightarrow \alpha$$

因为类型 `Fin n` 有且仅有 n 个取值，故若将其用作索引，则越界永不发生⁴²。

³⁷译注：也就是 `append [] v`。

³⁸译注：也就是 `append (Cons m v vs) w`。

³⁹译注：此段有译者大范围改写。原文为“Appending an empty vector to another vector v results in v . Appending a vector of the form `Cons m v vs` to a vector v works by invoking recursion via `rec` (which appends vs to w) and prepending v .”

⁴⁰译注：所谓全函数，就是对于任何输入，都有确定输出的函数。有些函数对于某些输入没有确定的输出（抛出异常、无限循环、引入未定义行为……），这样的函数相应地称为部分函数 (*partial function*)。

⁴¹译注：也就是下标索引：取向量/数组的第 n 个元素。

⁴²译注：此句为译者补文。

另一个值得注意的依值类型是相等类型 (*equality type*):

$$\mathbf{Eq} :: \forall (\alpha :: *) . \alpha \rightarrow \alpha \rightarrow *$$

它只有一个构造子:

$$\mathbf{Refl} :: \forall (\alpha :: *) (x :: \alpha) \rightarrow \mathbf{Eq} \alpha x x$$

而这是它的消去子:

$\mathbf{eqElim} :: \forall (\alpha :: *)$	元素类型
$\bullet \forall (m :: \forall (x :: \alpha) . \forall (y :: \alpha) . \mathbf{Eq} \alpha x y \rightarrow *)$	动机
$\bullet (\forall (z :: \alpha) . m z z (\mathbf{Refl} \alpha z))$	基础情况
$\rightarrow \forall (x :: \alpha) . \forall (y :: \alpha) . \forall (p :: \mathbf{Eq} \alpha x y)$	待消项
$\bullet m x y p$	返回类型

使用 \mathbf{Eq} , 我们可以直接在 λ_{Π} 中书写并证明与我们代码有关的定理。例如, 类型

$$\forall (\alpha :: *) (n :: \mathbf{Nat}) . \mathbf{Eq} \mathbf{Nat} (\mathbf{plus} \ n \ \mathbf{Zero}) \ n$$

表明 \mathbf{Zero} 是自然数加法中的右单位元 (*right identity*)⁴³。而根据柯里-霍华德同构 (*Curry-Howard isomorphism*), 任何具有以上类型的词项即是这一定理的证明。

这些例子和一些其他例子包含在论文源码附带的解释器中, 可以从 λ_{Π} 的主页 [7] 下载。关于适用于依值类型语言的数据类型, 以及如何编写依值类型程序的更多信息, 请参阅另一篇教程 [12]。

在本节中, 我们选择为新增的每种数据类型扩展核心语法。另一种选择是为数据类型使用邱奇编码 (*Church encoding*), 例如, 我们可以用类型 $\forall (\alpha :: *) . \alpha \rightarrow (\alpha \rightarrow \alpha) \rightarrow \alpha$ 来表示自然数。这一选择看似事半功倍, 实则后患无穷。我们能用品邱奇编码写出简单的 \mathbf{fold} , 却不能用它写出依赖于消去子的程序, 除非我们进一步扩展理论。这样一来, 要编写那些天生就依靠依值类型的程序——例如我们的 \mathbf{append} 函数——就变得举步维艰。而我们的核心理论应该能构成依值类型编程语言的坚实基础, 所以我们不使用这种编码方式。

5. 通往依值类型编程

我们现已描述的演算距离真正的程序设计语言尚且相距甚远。尽管我们已经可以写出简单的表达式、对其作类型检查并求值, 但距离它能被用来编写大型的复杂程序, 还有许多的工作要做。本节并不旨在列出进行大规模依值类型编程时所必须面对的全部问题, 更遑论解决这些问题。相反, 我们将尝试勾勒出如何在目前为止我们所看到的核心理论之上构建编程语言, 并推荐相关文献。

从上面的几个例子, 我们看得出来, 用消去子编程只是一时权宜, 并非长久之计。Epigram [15] 通过巧妙地选择动机, 使得用消去子编程变得更加实用 [9, 14]。通过选择合适的动机, 我们可以在定义复杂函数时利用类型信息。消去子看起来可能并不是那么有用, 但它们构成了依值类型编程语言得以建立的基础。

编写带有复杂类型的程序也绝非易事。Epigram 和 Agda [19] 允许程序员在代码里留下“空洞 (*hole*)”——让程序中的这些部分处于未定义的状态 [20]。程序员可以询问系统某个空洞具有什么类型, 这使得复杂程序能被有效率地逐步开发。

⁴³译注: “单位元 (*identity*)”原作 “neutral element”。

目前，我们的核心系统要求程序员显式地实例化多态函数，繁琐不堪。以我们定义的 `append` 函数为例，它的五个参数中，只有两个是重要的。幸运的是，不重要的参数通常可以被推断出来。许多基于依值类型的编程语言和证明助手都支持隐式参数 (*implicit arguments*)，用户可以在调用函数时省略这些参数。需要注意的是，这些参数不必是类型，例如 `append` 函数的“多态性”也可以体现在向量长度上。

最后，我们要重申我们现在所展示的类型系统是不健全 (*unsound*) 的。因为 `*` 本身的种类也是 `*`，我们可以编码出罗素悖论 (*Russell's paradox*) 的某种变体——吉拉德悖论 (*Girard's paradox*) [3]。这使得我们可以创造出具有任何类型的东西。要修复这个问题，标准的解决方案是为类型引入一个无限的层级：`*` 的类型是 `*1`，`*1` 的类型是 `*2`，依此类推。

6. 讨论

在类型论和类型系统的实现这方面有着大量的相关文献。Pierce 的书 [21] 是绝佳的起点之一，而 Martin L f 的类型论笔记 [8] 迄今为止仍有极高的价值，也是这一主题极佳的入门材料。Nordstr m 和 Thompson 等人近期所著的书籍 [17, 24] 也可在网络上免费取得。

目前已有若干依值类型编程语言和证明助手可供使用。Coq [2] 是一个成熟且文档齐全的证明助手。虽然它并非主要为依值类型编程而设计，但学习 Coq 有助于建立对类型论的直觉。Haskell 程序员可能会更适应较新版本的 Agda [18]——一种依值类型编程语言。Agda 的语法不仅与 Haskell 相似，而且还可以使用模式匹配和一般递归来定义函数。最后，Epigram [15, 12] 与我们所熟知的函数式编程进行了更彻底的决裂。尽管其初始实现远非完美，但 Epigram 的许多理念尚未在其他地方得到实现。

在引言中，我们提到了函数式程序员对依值类型的一些顾虑。对依值类型语言进行类型检查并不必然是不可判定的——事实上，我们在这里展示的类型检查器只会在某些刻意构造的例子 [3] 中无法停机。求值和类型检查之间的阶段区分变得更加微妙，但这一区分仍然存在。类型与词项的融合带来了新的挑战，但也提供了很多机会。但最重要的是，入门依值类型并不像你想象的那么难。我们希望这篇文章能够激发你的兴趣，引导你迈出第一步，并鼓励你自己开始探索依值类型！

致谢

我们要感谢 Thorsten Altenkirch、Lennart Augustsson、Isaac Dupree、Clemens Fruhwirth、Jurriaan Hage、Stefan Holdermans、穆信成、Edsko de Vries、Phil Wadler、乌特勒支大学的几期类型系统研讨班的学生们、Lambda the Ultimate 社区，以及匿名审稿人对本文早期版本提出的宝贵意见⁴⁴。

参考文献

- [1] Abel, A., Altenkirch, T.: A Partial Type Checking Algorithm for Type:Type, International Workshop on Mathematically Structured Functional Programming (V. Capretta, C. McBride, Eds.), 2008.
- [2] Bertot, Y., Cast eran, P.: Interactive Theorem Proving and Program Development. Coq'Art: The Calculus of Inductive Constructions, Springer Verlag, 2004.
- [3] Coquand, T.: An analysis of Girard's paradox, First IEEE Symposium on Logic in Computer Science, 1986.
- [4] Coquand, T.: An Algorithm for Type-Checking Dependent Types, Science of Computer Programming, 26(1-3), 1996, 167-177.
- [5] Coquand, T., Takeyama, M.: An Implementation of Type: Type, International Workshop on Types for Proofs and Programs, 2000.

⁴⁴译注：你们还要谢谢我，你们的译者。天知道我在翻译这论文的时候受了多少气——Chuigda Whitegive.

- [6] Hinze, R., Löh, A.: lhs2TEX, 2007, <http://www.cs.uu.nl/~andres/lhs2tex>.
- [7] $\lambda 5$ homepage, 2007, <http://www.cs.uu.nl/~andres/LambdaPi>.
- [8] Martin-Löf, P.: Intuitionistic type theory, Bibliopolis, 1984.
- [9] McBride, C.: Dependently Typed Functional Programs and their Proofs, Ph.D. Thesis, University of Edinburgh, 1999.
- [10] McBride, C.: Elimination with a Motive, TYPES '00: Selected papers from the International Workshop on Types for Proofs and Programs, Springer-Verlag, 2000.
- [11] McBride, C.: Faking it: Simulating Dependent Types in Haskell, Journal of Functional Programming, 12(5), 2002, 375-392.
- [12] McBride, C.: Epigram: Practical Programming with Dependent Types., Advanced Functional Programming, 2004.
- [13] McBride, C., McKinna, J.: Functional pearl: I am not a number - I am a free variable, Haskell '04: Proceedings of the 2004 ACM SIGPLAN workshop on Haskell, 2004.
- [14] McBride, C., McKinna, J.: The view from the left, Journal of Functional Programming, 14(1), 2004, 69-111.
- [15] McBride, C. et al.: Epigram, 2004, <http://www.e-pig.org>.
- [16] Meijer, E., Fokkinga, M., Paterson, R.: Functional Programming with Bananas, Lenses, Envelopes and Barbed Wire, 5th Conf. on Functional Programming Languages and Computer Architecture, 1991.
- [17] Nordström, B., Petersson, K., Smith, J. M.: Programming in Martin-Löf's Type Theory: An Introduction, Clarendon, 1990.
- [18] Norell, U.: Agda 2, <http://appserv.cs.chalmers.se/users/ulfn/wiki/agda.php>.
- [19] Norell, U.: Towards a practical programming language based on dependent type theory, Ph.D. Thesis, Chalmers University of Technology, 2007.
- [20] Norell, U., Coquand, C.: Type checking in the presence of meta-variables, Submitted to Typed Lambda Calculi and Applications 2007.
- [21] Pierce, B. C.: Types and Programming Languages, MIT Press, Cambridge, MA, USA, 2002, ISBN 0-262-16209-1.
- [22] Pierce, B. C., Turner, D. N.: Local Type Inference, ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages (POPL), San Diego, California, 1998, Full version in ACM Transactions on Programming Languages and Systems (TOPLAS), 22(1), January 2000, pp. 1-44.
- [23] Pollack, R.: Closure under alpha-conversion, TYPES '93: Proceedings of the international workshop on Types for proofs and programs, 1994.
- [24] Thompson, S.: Type Theory and Functional Programming, Addison Wesley Longman Publishing Co., Inc., 1991.